

お客様各位

カタログ等資料中の旧社名の扱いについて

2010年4月1日を以ってNECエレクトロニクス株式会社及び株式会社ルネサステクノロジが合併し、両社の全ての事業が当社に承継されております。従いまして、本資料中には旧社名での表記が残っておりますが、当社の資料として有効ですので、ご理解の程宜しくお願ひ申し上げます。

ルネサスエレクトロニクス ホームページ (<http://www.renesas.com>)

2010年4月1日

ルネサスエレクトロニクス株式会社

【発行】ルネサスエレクトロニクス株式会社 (<http://www.renesas.com>)

【問い合わせ先】 <http://japan.renesas.com/inquiry>

ご注意書き

1. 本資料に記載されている内容は本資料発行時点のものであり、予告なく変更することがあります。当社製品のご購入およびご使用にあたりましては、事前に当社営業窓口で最新の情報をご確認いただきますとともに、当社ホームページなどを通じて公開される情報に常にご注意ください。
2. 本資料に記載された当社製品および技術情報の使用に関連し発生した第三者の特許権、著作権その他の知的財産権の侵害等に関し、当社は、一切その責任を負いません。当社は、本資料に基づき当社または第三者の特許権、著作権その他の知的財産権を何ら許諾するものではありません。
3. 当社製品を改造、改変、複製等しないでください。
4. 本資料に記載された回路、ソフトウェアおよびこれらに関連する情報は、半導体製品の動作例、応用例を説明するものです。お客様の機器の設計において、回路、ソフトウェアおよびこれらに関連する情報を使用する場合には、お客様の責任において行ってください。これらの使用に起因しお客様または第三者に生じた損害に関し、当社は、一切その責任を負いません。
5. 輸出に際しては、「外国為替及び外国貿易法」その他輸出関連法令を遵守し、かかる法令の定めるところにより必要な手続を行ってください。本資料に記載されている当社製品および技術を大量破壊兵器の開発等の目的、軍事利用の目的その他軍事用途の目的で使用しないでください。また、当社製品および技術を国内外の法令および規則により製造・使用・販売を禁止されている機器に使用することができません。
6. 本資料に記載されている情報は、正確を期すため慎重に作成したのですが、誤りがないことを保証するものではありません。万一、本資料に記載されている情報の誤りに起因する損害がお客様に生じた場合においても、当社は、一切その責任を負いません。
7. 当社は、当社製品の品質水準を「標準水準」、「高品質水準」および「特定水準」に分類しております。また、各品質水準は、以下に示す用途に製品が使われることを意図しておりますので、当社製品の品質水準をご確認ください。お客様は、当社の文書による事前の承諾を得ることなく、「特定水準」に分類された用途に当社製品を使用することができません。また、お客様は、当社の文書による事前の承諾を得ることなく、意図されていない用途に当社製品を使用することができません。当社の文書による事前の承諾を得ることなく、「特定水準」に分類された用途または意図されていない用途に当社製品を使用したことによりお客様または第三者に生じた損害等に関し、当社は、一切その責任を負いません。なお、当社製品のデータ・シート、データ・ブック等の資料で特に品質水準の表示がない場合は、標準水準製品であることを表します。
標準水準： コンピュータ、OA 機器、通信機器、計測機器、AV 機器、家電、工作機械、パーソナル機器、産業用ロボット
高品質水準： 輸送機器（自動車、電車、船舶等）、交通用信号機器、防災・防犯装置、各種安全装置、生命維持を目的として設計されていない医療機器（厚生労働省定義の管理医療機器に相当）
特定水準： 航空機器、航空宇宙機器、海底中継機器、原子力制御システム、生命維持のための医療機器（生命維持装置、人体に埋め込み使用するもの、治療行為（患部切り出し等）を行うもの、その他直接人命に影響を与えるもの）（厚生労働省定義の高度管理医療機器に相当）またはシステム等
8. 本資料に記載された当社製品のご使用につき、特に、最大定格、動作電源電圧範囲、放熱特性、実装条件その他諸条件につきましては、当社保証範囲内でご使用ください。当社保証範囲を超えて当社製品をご使用された場合の故障および事故につきましては、当社は、一切その責任を負いません。
9. 当社は、当社製品の品質および信頼性の向上に努めておりますが、半導体製品はある確率で故障が発生したり、使用条件によっては誤動作したりする場合があります。また、当社製品は耐放射線設計については行っておりません。当社製品の故障または誤動作が生じた場合も、人身事故、火災事故、社会的損害などを生じさせないようお客様の責任において冗長設計、延焼対策設計、誤動作防止設計等の安全設計およびエージング処理等、機器またはシステムとしての出荷保証をお願いいたします。特に、マイコンソフトウェアは、単独での検証は困難なため、お客様が製造された最終の機器・システムとしての安全検証をお願いいたします。
10. 当社製品の環境適合性等、詳細につきましては製品個別に必ず当社営業窓口までお問合せください。ご使用に際しては、特定の物質の含有・使用を規制する RoHS 指令等、適用される環境関連法令を十分調査のうえ、かかる法令に適合するようご使用ください。お客様がかかる法令を遵守しないことにより生じた損害に関し、当社は、一切その責任を負いません。
11. 本資料の全部または一部を当社の文書による事前の承諾を得ることなく転載または複製することを固くお断りいたします。
12. 本資料に関する詳細についてのお問い合わせその他お気付きの点等がございましたら当社営業窓口までご照会ください。

注 1. 本資料において使用されている「当社」とは、ルネサスエレクトロニクス株式会社およびルネサスエレクトロニクス株式会社とその総株主の議決権の過半数を直接または間接に保有する会社をいいます。

注 2. 本資料において使用されている「当社製品」とは、注 1 において定義された当社の開発、製造製品をいいます。

M16C/60 シリーズ クロック同期形シリアル I/O

1. 要約

この資料は M16C/60 シリーズクロック同期形シリアル I/O の使用方法を紹介し、応用例を掲載しています。

2. はじめに

この資料で説明する応用例は、次のマイコンでの利用に適用されます。

・マイコン：M16C/60 シリーズ

3.0 応用例の説明

3.1 クロック同期形シリアル I/O 概要

クロック同期形シリアル I/O は、1本のクロック、1本のデータ線（双方向の場合は2本）、および1～2本の制御線（場合により異なる）を使用し、1ビットずつ同期を取ってデータを伝送する方式です。パラレル伝送に比べ、少ない配線本数で伝送が行えます。




図 1. クロック同期形シリアルインターフェース例

3.2 M16C/60 シリーズのクロック同期形シリアル I/O の特長

M16C/60 シリーズのクロック同期形シリアル I/O は、様々な外部デバイスへの対応を可能にするための、またソフトウェア開発を容易にするための機能を持っています。

注. すべてのシリアル I/O が下記の機能をすべて持っているわけではありません。また、機能によっては、他の機能と同時に使用できない場合があります。

・転送クロックに対するデータ入出力タイミングを選択可

送受信データの出力・入力を、転送クロックの立ち上がりで行うか、立ち下がりで行うかを選択できます。

・データ入出力順序を選択可

送受信データの出力・入力を、ビット 0 から行うか (LSB ファースト)、ビット 7 から行うか (MSB ファースト) を選択できます。

・入出力データの論理切り替え可 (一部のシリアル I/O のみ)

送信データの書き込み、受信データの読み出しの際、データを反転できます。

・データ入出力端子の極性切り替え可 (一部のシリアル I/O のみ)

送信データ出力端子、受信データ入力端子のレベルを反転できます。

・割り込み要因を選択可

送信割り込み要求が発生するタイミングを、送信バッファレジスタが空になったとき、または送信完了時から選択できます。

・CTS/RTS機能を選択可

データ送信時に CTS 機能を、データ受信時に RTS 機能を使用できます。

・受信時のエラー検出機能

データ受信時、データの読み飛ばしが発生した場合に、フラグによりエラー検出できます (オーバランエラー)。

・転送クロックを複数端子から出力可 (一部のシリアル I/O のみ)

転送クロック出力端子を 2 本持ち、2 本を切り替えて使用できるため、2 つのデバイスに対し、データ線共通で送受信を行えます。

・連続受信モード

データ受信ごとに送信バッファレジスタにダミーデータを書き込んで次の受信を許可する、通常の受信方法のほかに、

1 回のダミーデータ書き込み後は、受信バッファレジスタから受信データを読み出すことにより次の受信を許可する、連続受信モードを選択できます。

表 1. グループ/チャンネルごとのクロック同期形シリアル I/O の機能比較

機能 / グループ チャンネル	M16C/61			M16C/62				
	クロック同期/非同期			クロック同期/非同期			クロック同期	
	UART0	UART1	UART2	UART0	UART1	UART2	SIO3	SIO4
CLK 極性選択	○	○	○	○	○	○	×	×
LSB ファースト/ MSB ファースト選択	○	○	○	○	○	○	○	○
転送クロック 複数端子出力機能	×	○	×	×	○	×	×	×
CTS/RTS 機能	○	○	○	○	○	○	×	×
シリアルデータ 論理切り替え	×	×	○	×	×	○	×	×
TxD, RxD 入出力 極性切り替え	×	×	○	×	×	○	×	×
連続受信モード	○	○	○	○	○	○	×	×
送信割り込み タイミング選択	○	○	○	○	○	○	×	×

注.クロック同期専用の 2 本 (SIO3、SIO4) とクロック同期/非同期兼用の 3 本 (UART0~2) では、設定方法、動作などに多少の違いがあります。

[相違点]

- クロック同期専用の 2 本では、転送データを書き込むことにより、送受信が開始します。クロック同期/非同期兼用の 3 本では、転送データの書き込み前に、送信許可ビット、受信許可ビットの設定が必要です。
- 送信完了・受信完了は、クロック同期専用の 2 本では、割り込み要求ビットのみで判定できます。クロック同期/非同期兼用の 3 本では、割り込み要求ビット、および送信レジスタ空フラグ、受信完了フラグで判定できます。
- 連続して送受信を行う場合、クロック同期/非同期兼用の 3 本では、送信バッファレジスタまたは受信レジスタが空になると、次の送信データの送信または次のデータの受信が可能です。クロック同期専用の 2 本では、直前の送受信が完了するまで、次の送信データの送信 (次のデータの受信) は行えません。
(UART0~2 は、送信用・受信用にそれぞれ 2 つずつレジスタを持つ W バッファであるのに対し、SIO3、SIO4 は、それぞれ 1 つしかレジスタを持たないため)

3.3 クロック同期形シリアル I/O の動作

M16C/60 シリーズのクロック同期形シリアル I/O では、各制御レジスタの設定により、モード選択、転送速度設定、送受信許可を行えます。送受信は、送信時には送信データを、受信時にはダミーデータを送信バッファレジスタに書き込むことにより、自動的に開始します。送受信の完了は、制御レジスタ中の送信レジスタ空フラグ、受信完了フラグにより判定できます。

内部クロック/外部クロック選択

転送クロックを外部に出力する場合は、内部クロックを選択します。後述の転送速度設定を行い、送受信を許可した後、送信バッファレジスタに送信データ（送信時）またはダミーデータ（受信時）を書き込むと、送受信が開始します。

転送クロックを外部から入力する場合は、外部クロックを選択します。送受信を許可した後、送信バッファレジスタに送信データ（送信時）またはダミーデータ（受信時）を書き込むことにより、外部クロックの入力準備が完了します。

転送速度設定

内部クロック選択時は、転送速度を選択できます。転送速度は、転送速度レジスタのカウントソース（f1、f8、f32）、および設定値（0～255）により決定されます。（転送速度レジスタの出力は、更に 1/2 分周（固定）されて転送クロックになります。）




図 2. 転送速度設定

転送速度条件

転送速度は、ユーザーズマニュアルの電気的特性に記載しているシリアル I/O のタイミング必要条件を満たす必要があります。

表 2.M16C/62 グループ シリアル I/O タイミング必要条件 (Vcc=5V)

記号	項目	規格値		単位
		最小	最大	
t _c (CK)	CLKi 入力サイクル時間	200		ns
t _w (CKH)	CLKi 入力“H”パルス幅	100		ns
t _w (CKL)	CLKi 入力“L”パルス幅	100		ns
t _d (C-Q)	TxDi 出力遅延時間		80	ns
t _h (C-Q)	TxDi ホールド時間	0		ns
t _{su} (D-C)	RxDi 入力セットアップ時間	30		ns
t _h (C-D)	RxDi 入力ホールド時間	90		ns

例えば、Vcc=5V、f(XIN)=16MHz 時、転送速度は、計算式 $f_i/2(n+1)$ においては、最速で 8Mbps となります (fi=f1 かつ n=0 の場合)。しかし、タイミング必要条件で規定されるクロックの最小入力サイクル時間は 200ns であり、計算式上の最速値 8Mbps はこの値を満たしていません。この場合、最速値は、n=1 のときで、4Mbps になります。

また、接続デバイスでは、送信時は TxD の出力遅延最大時間、受信時は RxD の入力セットアップ最小時間を考慮してください。

送受信開始

送信時は、送信許可ビットをセットした後、送信データを送信バッファレジスタに書き込みます。受信時は、送信許可ビット、受信許可ビットの双方をセットした後、ダミーデータを送信バッファレジスタに書き込みます。

注. M16C/62 グループの TxD2 端子は、N チャネルオープンドレイン出力であるためプルアップが必要です。




図 3. フローチャート (送信時)




図 4. フローチャート (受信時)

3.4 割り込みを使用した連続データ送信

連続したデータを、割り込み処理内で設定、送信する場合、送信割り込み要因を「送信バッファレジスタが空になったとき」に設定することにより、データーデータ間の空白時間を最小にすることができます。

送信バッファレジスタに書き込まれた 8 ビットデータは、送信レジスタに転送された後、TxD 端子より出力されます。送信割り込み要因選択ビットで割り込み要因を「送信完了」に設定した場合、割り込み要求は 8 ビットデータの出力が完了した時点で発生します。これに対し「送信バッファ空」に設定した場合は、送信バッファレジスタから送信レジスタにデータが転送された時点で割り込み要求が発生し、送信が完了する前に、次のデータを送信バッファレジスタに設定することができます。




図 5. 割り込み要因に「送信バッファレジスタが空になったとき」を選択した場合の動作タイミング図




図 6. 割り込み要因に「送信完了時」を選択した場合の動作タイミング図

割り込みを使用した送信処理例

メインルーチンで送信を許可し、割り込み処理でデータを送信する

【概要】

1. あらかじめ、クロック同期形シリアル I/O の各機能の設定を行います。
2. 使用する送信割り込み要求ビットをクリアした後、割り込み許可フラグをセットします。
3. 送信開始時に、送信許可ビットをセットし、1 バイト目の送信データを送信バッファレジスタに書き込みます。
4. 割り込み発生ごとに、送信データを 1 バイトずつ送信バッファレジスタに書き込みます。
5. 最終データを書き込んだ後、送信割り込みを禁止します。




図 7. 概略フローチャート

サンプルプログラム
500h 番地～503h 番地のデータを送信

```

      .
      .
      .
;===== メインプログラム =====
MAIN:
  JSR      INIT          ;UART0の初期設定へ
  FSET    I             ;割り込み許可
      .
      .
  MOV.B   #01h,UOC1     ;送信許可
  MOV.B   [A0],UOTB     ;送信データの書き込み
      .
      .
;===== UART0機能初期設定 =====
INIT:
  MOV.B   #01h,UOMR     ;内部クロック選択
  MOV.B   #10h,UOCO     ;CTS機能禁止、カウントsf1を選択
  MOV.B   #00h,UCON     ;UART0送信割り込み要因=送信バッファ空
  MOV.B   #0FFh,UOBRG   ;転送速度=f1/256×1/2
  MOV.W   #500h,DATA    ;送信データ取り出し位置の指定
  MOV.W   #500h,A0
  MOV.B   #3,CNT        ;転送回数の設定(4-1)
  MOV.B   #01h,SOTIC    ;UART0送信割り込み優先レベルの設定
INT_END:
  RTS
      .
      .
;===== UART0送信割り込み処理 =====
SOT_INT:
  PUSH.W  A0            ;A0を待避
  INC.B   DATA        ;次の送信データの取り出し位置の指定
  MOV.W   DATA,A0
  MOV.B   [A0],UOTB    ;次の送信データの書き込み
  DEC.B   CNT          ;転送回数カウントダウン
  JNZ    SOT_INT_1    ;次の送信データがあるときは
                  ;割り込み許可のまま
  MOV.B   #00h,SOTIC   ;最終送信データの場合,送信割り込み禁止
SOT_INT_1:
  POP.W   A0           ;A0を復帰
  REIT
      .
      .
      .

```

3.5 連続受信モード

連続受信モードでは、送信バッファレジスタにダミーデータを書き込むことなく、連続してデータを受信できます。（受信開始のために、最初の 1 回は、送信バッファレジスタへのダミーデータの書き込み、または受信バッファレジスタからのダミーデータの読み出しが必要です。）

通常の受信方法では、1 バイトデータを受信する度に、送信バッファレジスタにダミーデータを書き込むことによって、次のデータを受信が可能になります。連続受信モードでは、受信バッファレジスタからの受信データの読み出しによって次のデータを受信が可能になり、送信バッファレジスタへの書き込みは不要です。（受信バッファレジスタを読み出すことにより、自動的に送信バッファレジスタにダミーデータが書き込まれます。）

注 1. 連続受信モード時は、最初の 1 回を除いて、送信バッファレジスタへのダミーデータの書き込みは行わないでください。

注 2. TxD 端子から出力される値は不定です。




図 8. 受信動作比較

連続受信モード時の処理例

メインルーチンで受信を許可し、割り込み処理でデータを読み出す

[概要]

1. あらかじめ、連続受信モードを許可し、クロック同期形シリアル I/O の各機能の設定を行います。
2. 使用する受信割り込み要求ビットをクリアした後、割り込み許可フラグをセットします。
3. 受信開始時に、送信許可ビット、受信許可ビットをセットし、受信バッファレジスタからダミーデータを読み出します。
4. 割り込み発生ごとに、受信データを受信バッファレジスタから読み出します。
5. 最終データを読み出した後、受信割り込みを禁止します。




図 9. 概略フローチャート

サンプルプログラム
受信データを 500h 番地～503h 番地に格納

```

      .
      .
;===== メインプログラム =====
MAIN:
  JSR      INIT          ;UART0の初期設定へ
  FSET     I             ;割り込み許可
      .
  MOV.B   #05h,UOC1     ;受信許可
  MOV.B   UORB,ROL      ;データ読み出し=受信開始
      .
;===== UART0機能初期設定 =====
INIT:
  MOV.B   #09h,UOMR     ;外部クロック選択
  MOV.B   #10h,UOCO     ;RTS機能禁止
  MOV.B   #04h,UCON     ;UART0連続受信モード許可
  MOV.W   #500h,DATA    ;受信データ格納位置の指定
  MOV.B   #4,CNT        ;転送回数の設定
  MOV.B   #01h,SORIC    ;UART0受信割り込み優先レベルの設定
INT_END:
  RTS
      .
;===== UART0受信割り込み処理 =====
SOR_INT:
  PUSH.W  A0            ;A0を待避
  MOV.W   DATA,A0
  MOV.B   UORB,[A0]     ;受信データの読み出し
  INC.B   DATA         ;次の受信データの格納位置の指定
  DEC.B   CNT           ;転送回数カウンタ減
  JNZ     SOR_INT_1     ;次の受信データがあるときは
                       ;割り込み許可のまま
  MOV.B   #00h,SORIC    ;最終受信データの場合、受信割り込み禁止
SOR_INT_1:
  POP.W   A0            ;A0を復帰
      .
      .

```

3.6 転送クロック複数端子出力機能

転送クロック複数端子出力機能を選択し、2本の転送クロック出力端子を切り替えて使用すると、2つの外部ICに対し、データ線共通で、時分割でデータ送受信を行うことができ、使用する端子数や基板上の配線を少なくすることができます。




図 10. 接続図（内部クロックで送信のみ行う場合）




図 11. 動作タイミング図

転送クロック複数端子出力機能使用時の処理例

- (1) メインルーチンで送信を許可し、転送クロック出力端子として CLK_{S1} を選択する
- (2) 割り込みが発生する度に、転送クロック出力端子を切り替え (→ CLK₁ → CLK_{S1} → CLK₁...)、2 つの外部 IC に対して交互にデータを送信する

【概要】

1. あらかじめ、転送クロック複数端子出力機能を選択し、クロック同期形シリアル I/O の各機能の設定を行います。
2. 使用する送信割り込み要求ビットをクリアした後、割り込み許可フラグをセットします。
3. 送信開始時に、送信許可ビットをセットし、1 バイト目の送信データを送信バッファレジスタに書き込みます。(最初のデータは CLK_{S1} 側に送信します。)
4. 割り込み発生ごとに、CLK₁、CLK_{S1} を切り替え、送信データを 1 バイトずつ送信バッファレジスタに書き込みます。
5. 最終データを書き込んだ後、送信割り込みを禁止します。




図 12. 概略フローチャート

サンプルプログラム
500h 番地~503h 番地のデータを送信

```

      .
      .
      .
;----- メインプログラム -----
MAIN:
  JSR     INIT           ;UART1の初期設定へ
  FSET   I              ;割り込み許可
      .
      .
  MOV.B  #01h,U1C1      ;送信許可
  MOV.B  [A0],U1TB      ;送信データの書き込み
      .
      .
;----- UART1機能初期設定 -----
INIT:
  MOV.B  #01h,U1MR      ;内部クロック選択
  MOV.B  #10h,U1C0      ;CTS機能禁止、カウントスf1選択
  MOV.B  #32h,UCON      ;転送クロック複数端子出力機能選択、
                        ;クロック出力端子CLKS1に設定
  MOV.B  #0FFh,U1BRG    ;転送速度=f1/256×1/2
  MOV.W  #500h,DATA     ;送信データ取り出し位置の指定
  MOV.W  #500h,A0
  MOV.B  #3,CNT         ;転送回数の設定(4-1)
  MOV.B  #01h,S1TIC     ;送信割り込み許可
INT_END:
  RES
      .
      .
;----- UART1送信割り込み処理 -----
S1T_INT:
  PUSH.W A0             ;A0を待避
  XOR.B  #10h,UCON      ;クロック出力端子(CLK1,CLKS1)の切り替え
  INC.B  DATA          ;次の送信データの取り出し位置の指定
  MOV.W  DATA,A0
  MOV.B  [A0],U1TB      ;次の送信データの書き込み
  DEC.B  CNT             ;転送回数カウンタ減
  JNZ   S1T_INT_1       ;次の送信データがあるときは
                        ;割り込み許可のまま
                        ;最終送信データの場合、送信割り込みの禁止
  MOV.B  #00H,S1TIC
S1T_INT_1:
  POP.W  A0              ;A0を復帰
  REIT
      .
      .
      .

```

3.7 CTS/RTS 機能

CTS (Clear To Send) 機能は、外部 IC の状態を検知し、これに応じて送信を制御する機能です。この機能を選択した場合、CTS 端子の入力が"L"レベルになると、自動的に転送クロックが発生し、送信を開始します。

CTS 端子のレベルの確認は送信開始時に行われるため、送信中に CTS 端子が"H"レベルになった場合は、実行中の送信が完了した後、転送クロックが停止します。また、その後、CTS 端子が"L"レベルになると、再度、転送クロックが発生します。

RTS (Request To Send) 機能は、受信可能状態を外部 IC に知らせる機能です。この機能を選択した場合、受信準備が完了した時点で、自動的に RTS 端子から"L"レベルが出力されます。その後、RTS 端子は、転送クロックの最初の立ち上がりで、"H"レベルになります。

CTS/RTS 機能については、CTS 機能のみ有効、RTS 機能のみ有効、両機能禁止のいずれかを選択することができます。

※両機能禁止時、CTS/RTS 端子はプログラマブル入出力ポートとして使用できます。




図 13. CTS 機能使用時の接続例





図 14. 動作タイミング図



注. M16C/62 グループの TxD2 端子は、
 N チャネルオープンドレイン出力であるためプルアップが必要です。

図 15. RTS 機能使用時の接続例




図 16. 動作タイミング図

RTS 機能使用時の処理例

メインルーチンで受信を許可し、割り込み処理でデータを読み出す

[概要]

1. あらかじめ、RTS 機能を選択し、クロック同期形シリアル I/O の各機能の設定を行います。
2. 使用する受信割り込み要求ビットをクリアした後、割り込み許可フラグをセットします。
3. 送信許可ビット、受信許可ビットをセットし、送信バッファレジスタにダミーデータを書き込むと、RTS 端子の出力レベルが"L"になり、送信側に受信可能状態であることを知らせます。
4. 割り込み発生ごとに、受信データを受信バッファレジスタから読み出します。
5. 最終データを受信するまで、次のデータ受信のため、送信バッファレジスタにダミーデータを書き込みます。
6. 最終データを読み出した後、受信割り込みを禁止します。




図 17. 概略フローチャート

サンプルプログラム

受信データを 500h 番地～503h 番地に格納

```

      .
      .
      .
;===== メインプログラム =====
MAIN:
  JSR      INIT          ;UART0の初期設定へ
  FSET     I            ;割り込み許可
      .
      .
  MOV.B   #05h,U0C1     ;受信許可
  MOV.B   ROL,UORB      ;ダミーデータの書き込み=受信開始
      .
      .
;===== UART0機能初期設定 =====
INIT:
  MOV.B   #09h,U0MR     ;外部クロック選択
  MOV.B   #04h,U0C0     ;RTS機能許可
  MOV.B   #00h,U0CON    ;UART0連続受信モード禁止
  MOV.W   #500h,DATA    ;受信データ格納位置の指定
  MOV.B   #4,CNT        ;転送回数の設定
  MOV.B   #01h,SORIC    ;UART0受信割り込み優先レベルの設定
INT_END:
  RTS
      .
      .
;===== UART0受信割り込み処理 =====
SOR_INT:
  PUSH.W  A0            ;A0を待避
  PUSH.B  ROL           ;ROLを待避
  MOV.W   DATA,A0
  MOV.B   UORB,[A0]     ;受信データの読み出し
  INC.B   DATA         ;次の受信データの格納位置の指定
  DEC.B   CNT           ;転送回数カウントダウン
  JNZ     SOR_INT_1     ;次の受信データがあるときは
                      ;割り込み許可のまま
  MOV.B   #00h,SORIC    ;最終受信データの場合、受信割り込み禁止
SOR_INT_1:
  POP.B   ROL           ;ROLを復帰
  POP.W   A0            ;A0レジスタの復帰
  REIT
      .
      .
      .

```

3.8 エラー発生時の復帰方法

クロック同期形シリアル I/O 使用時、エラーが発生した場合は、初期化などの回復処理が必要となります。また、エラー状態からの復帰後、必要に応じて、ソフトウェアで再送処理や再送要求処理を行ってください。

受信エラー検出

M16C/60 シリーズのクロック同期形シリアル I/O では、オーバランエラーのみを検出することができます。オーバランエラーは、受信バッファレジスタのデータを読み出す前に、受信レジスタに次の受信データが揃ったときに発生します。このエラーが発生した場合、受信バッファレジスタには最後に受信したデータが格納されます。また、受信割り込み要求ビットは"1"になりません。このエラーが発生すると、オーバランエラーフラグが"1"になります。

エラーからの復帰

エラー状態からの復帰のための回復処理は、下記フローチャートに従って、実行してください。設定順序を変えると正常に動作しない場合があります。

また、オーバランエラー以外のエラーの検出には、ソフトウェアにて対応してください。

表 3. エラー処理

エラー原因	判定方法	復帰方法
オーバランエラー	エラーフラグ参照	初期化 再送要求処理
ビットずれなどによる 通信未完了	ソフトウェアによる タイムアウト判定	初期化 再送処理/再送要求処理
その他、 異常データなど	ソフトウェアによる データ判定など	再送処理/再送要求処理




図 18. 回復処理（初期化など）フローチャート（1）




図 19. 回復処理（初期化など）フローチャート（2）

3.9 I²Cモードを使用したEEPROMとの通信

I²Cバスは、PHILIPS社が開発した双方向マルチ・マスタ・バスの通信プロトコルで、多くのICで採用されています。(I²Cバス通信プロトコルの詳細については、PHILIPS社のI²Cバス仕様書をご参照ください。) M16C/60 シリーズでは、UART2 に簡易I²Cモードを内蔵し、ソフトウェアと組合せることによって、I²Cバスインタフェース (シングルマスタ/マルチマスタ) を実現できます。




図 20. 接続例

I²Cバスの概要

I²Cバスは、データ (SDA) とクロック (SCL) を使用する 2 線シリアルインタフェースで、これらの信号のレベルや変化点に意味を持たせて、通信を行います。通信順序は、マスタによる開始条件 (スタートコンディション) の送付、マスタまたはスレーブによるデータ送付/受信、マスタによる停止条件 (ストップコンディション) の送付、となります。




図 21. I²Cバスタイミングチャート

I²Cバスにおけるリード/ライト時のフォーマットは、開始条件、7 ビットのスレーブアドレス、1 ビットのリード/ライト設定、8 ビット単位のデータ、最後に停止条件、となります。また、この間、8 ビットデータごとに、データ受信側が確認応答 (アクノリッジ: SDA="L") を返します。




図 22. I²Cバス リード/ライトフォーマット

簡易I²Cモード時の処理例

「M16C/62 グループ アプリケーションノート 簡易I²Cバス編」に、I²Cバス通信プロトコルの概要、およびマルチマスタ通信方法を記載していますので、ご参照ください。

M16C/62 グループをシングルマスタとし、I²Cバスインタフェースを持つEEPROMと通信を行う場合のサンプルプログラムを以下に示します。

(リード/ライトフォーマットは一例です。)

ライトモード時の処理例

CPU (マスタ) から EEPROM (スレーブ) に 1 バイトのデータを書き込む

[概要]




図 23. タイミングチャート (ライトモード時)

1. あらかじめ、簡易I²Cモードを選択し、クロック同期形シリアルI/Oの各機能の設定を行います。
2. 開始条件 (スタートコンディション) を送出します。
3. スレーブアドレスを出力し、アクノリッジをチェックします。
4. ワード (書き込み先) アドレスを出力し、アクノリッジをチェックします。
5. データを出力し、アクノリッジをチェックします。
6. 停止条件 (ストップコンディション) を送出します。

注. アクノリッジのチェック時、NACK (アクノリッジ未検出: SDA="H") であった場合は、停止条件 (ストップコンディション) 送出後、再書き込み処理などのソフトウェアによる対応が必要です。

サンプルプログラム

```

*****
;
; Copyright 2003 Renesas Technology America, Inc.
; AND Renesas Solutions Corporation
; All rights reserved
;===== メインプログラム =====
;
;
MAIN:
    JSR    INIT          ;機能初期設定
    JSR    MASTER_OUT   ;ライトモード処理

*****
;
;                               機能初期設定
*****
;
;
INIT:
    BCLR   PD7_0         ;P7_0 入力
    BSET   P7_0         ;P7_0='H'
    BCLR   PD7_1         ;P7_1 入力
    BSET   P7_1         ;P7_1='H'
;
;
    MOV.B  #01h,U2SMR    ;IIC モード選択
    MOV.B  #82h,U2SMR2   ;アクノリッジ検出/アクノリッジ未検出割込有効
                        ;SCL 同期化機能有効
;
    MOV.B  #40h,U2SMR3   ;
    MOV.B  #00h,U2MR     ;シリアル I/O 無効=ポート制御
    MOV.B  #90h,U2C0     ;MSB ファースト、CTS/RTS 機能禁止
    MOV.B  #10h,U2C1     ;送受信禁止
    MOV.B  #(80-1),U2BRG ;転送速度 100Kbps
INIT_END:
    RTS

*****
;
;                               ライトモード処理
*****
;
;
MASTER_OUT:
    JSR    MAKE_START    ;スタート条件生成
    MOV.B  #0A0h,R0L     ;スレーブアドレス+データ方向ビット(W)
    JSR    OUTPUT_DATA   ;スレーブアドレス+データ方向ビット(W)送信
    JNC   MASTER_OUT_10 ;C=1:アクノリッジ未検出
    JSR   ABORT          ;エラー処理へ
    FSET  C              ;C=1->再書き込み処理などへ
    RTS

MASTER_OUT_10:
    MOV.B  #80h,R0L     ;ワードアドレス
    JSR    OUTPUT_DATA   ;ワードアドレス送信
    JNC   MASTER_OUT_11 ;C=1:アクノリッジ未検出
    JSR   ABORT          ;エラー処理へ
    FSET  C              ;C=1->再書き込み処理などへ
    RTS

```


MASTER_OUT_11:

```

MOV.B #55h,R0L ;書き込みデータ (55H)
JSR OUTPUT_DATA ;書き込みデータ送出
JNC MASTER_OUT_12 ;C=1:アクノリッジ未検出
JSR ABORT ;エラー処理へ
RTS

```

MASTER_OUT_12:

```

BTST P7_1
JNC MASTER_OUT_12 ;SCL(P7_1)が'L'の間待つ
BCLR TE_U2C1 ;送信禁止
JSR MAKE_STOP ;停止条件生成
RTS

```

```

;*****
;
; 開始条件(スタートコンディション)生成
;*****
;
;

```

MAKE_START:

```

JSR WAIT_5microSEC
JSR CHG_scl_H ;SCL='H'

```

MAKE_ST1:

```

BTST P7_1
JNC MAKE_ST1 ;SCL(P7_1)端子の入力が'H'になるまで待つ
BTST P7_0
JC MAKE_ST2 ;SDA(P7_0)='H'->MAKE_ST2
JSR CHG_sda_L ;SDA(P7_0)から'L'出力
MOV.B #00h,U2MR ;シリアル I/O 無効=ポート制御
JSR CHG_scl_L ;SCL(P7_1)から'L'出力
JSR WAIT_10microSEC
JSR CHG_sda_H ;SDA='H'
JSR WAIT_10microSEC

```

MAKE_ST2:

```

JSR CHG_sda_L ;開始条件送出
JSR WAIT_5microSEC
MOV.B #02h,U2MR ;IIC モード設定
RTS

```

```

;*****
;
; 停止条件(ストップコンディション)生成
;*****
;
;

```

MAKE_STOP:

```

BCLR TE_U2C1 ;送信禁止
BCLR RE_U2C1 ;受信禁止
BCLR PD7_0 ;SDA(P7_0)入力
MOV.B #00h,U2MR ;シリアル I/O 無効=ポート制御
JSR CHG_scl_L ;SCL(P7_1)'L'出力
JSR CHG_sda_L ;SDA(P7_0)'L'出力
JSR WAIT_5microSEC
JSR CHG_scl_H ;SCL(P7_1)='H'

```

MAKE_SP_1:

```

BTST P7_1
JNC MAKE_SP_1 ;SCL(P7_1)が'H'になるまで待つ

```

```
JSR    WAIT_10microSEC
JSR    CHG_sda_H          ;停止条件送出
RTS
```

```
*****
;
;                      データ送信
;
*****
```

```
OUTPUT_DATA:
    BSET    TE_U2C1          ;送信許可
    BCLR    IR_S2TIC        ;アクノリッジ未検出割り込み要求フラグクリア
    BCLR    IR_S2RIC        ;アクノリッジ検出割り込み要求フラグクリア
    OR.W    #100h,R0        ;SDA 開放ビット('H')付加
    MOV.W   R0,U2TB         ;送信データ設定

OUTPUT_1:
    BTST    IR_S2RIC        ;アクノリッジ検出割り込み要求フラグ=1?
    JNC     OUTPUT_2        ;アクノリッジ未検出割り込み要求フラグの判定へ
    JSR     WAIT_10microSEC
    FCLR    C
    RTS          ;C=0 のときアクノリッジ検出
```

```
OUTPUT_2:
    BTST    IR_S2TIC        ;アクノリッジ未検出割り込み要求フラグ=1?
    JNC     OUTPUT_1        ;アクノリッジ検出割り込み要求フラグの判定へ
    RTS          ;C=1 のときアクノリッジ未検出
```

```
*****
;
;                      停止処理(NACK 時)
;
*****
```

```
ABORT:
    BCLR    PD7_0
    BSET    P7_0            ;SDA(P7_0)='H'
    BCLR    TE_U2C1        ;送信禁止
    MOV.B   #00h,U2MR      ;シリアル I/O 無効=ポート制御
    JSR     WAIT_5microSEC
    JSR     MAKE_STOP      ;ストップ信号送出
    RTS
```

```
*****
;
;                      ポート切り替え処理
;
*****
```

```
CHG_scl_H:          ;SCL='H'
    BCLR    PD7_1
    RTS

CHG_scl_L:          ;SCL から'L'出力
    BCLR    P7_1
    NOP
    BSET    PD7_1
    RTS
```

```
CHG_sda_H:          ;SDA='H'
```

```
BCLR PD7_0
RTS
```

```
CHG_sda_L: ;SDA から'L'出力
    BCLR P7_0
    NOP
    BSET PD7_0
    RTS
```

```
*****
;
;                               待ち時間処理
;
*****
```

```
WAIT_10microSEC:
    .MREPEAT 40
    NOP
    .ENDR
```

```
WAIT_7_5_microSEC:
    .MREPEAT 40
    NOP
    .ENDR
```

```
WAIT_5microSEC:
    .MREPEAT 40
    NOP
    .ENDR
```

```
WAIT_2_5microSEC:
    .MREPEAT 40
    NOP
    .ENDR
    RTS
```

リードモード時の処理例

CPU (マスタ) が EEPROM (スレーブ) から 1 バイトのデータを読み出す

[概要]




図 24. タイミングチャート (リードモード時)

1. あらかじめ、簡易I²Cモードを選択し、クロック同期形シリアルI/Oの各機能の設定を行います。
2. 開始条件 (スタートコンディション) を送出します。
3. スレーブアドレスを出力し、アクノリッジをチェックします。
4. ワード (読み出し先) アドレスを出力し、アクノリッジをチェックします。
5. 開始条件 (スタートコンディション) を送出します。
6. スレーブアドレスを出力し、アクノリッジをチェックします。
7. 送出されたデータを取り込み、最後に **NACK** を出力します。
8. 停止条件 (ストップコンディション) を送出します。

注. アクノリッジのチェック時、**NACK** (アクノリッジ未検出 : SDA="H") であった場合は、停止条件 (ストップコンディション) 送出後、再書き込み処理などのソフトウェアによる対応が必要です。

サンプルプログラム

```

*****
;
; Copyright 2003 Renesas Technology America, Inc.
; AND Renesas Solutions Corporation
; All rights reserved
***** ワーク RAM 領域確保 *****
;
;
;          .SECTION    WORK,DATA
;          .ORG        00400H
WORKRAM_TOP:
DATA:      .BLKB      1      ;受信データ格納
WORKRAM_END:
;
;
;===== メインプログラム =====
;
;
MAIN:
        JSR    INIT      ;機能初期設定
        JSR    MASTER_IN ;リードモード処理

*****
;
;          機能初期設定
*****
;
;
INIT:
        BCLR   PD7_0      ;P7_0 入力
        BSET   P7_0       ;P7_0='H'
        BCLR   PD7_1      ;P7_1 入力
        BSET   P7_1       ;P7_1='H'
;
;
        MOV.B  #01h,U2SMR ;IIC モード選択
        MOV.B  #82h,U2SMR2 ;アクノリッジ検出/アクノリッジ未検出割込有効
;SCL 同期化機能有効
;
        MOV.B  #40h,U2SMR3 ;
        MOV.B  #00h,U2MR   ;シリアル I/O 無効=ポート制御
        MOV.B  #90h,U2C0   ;MSB ファースト、CTS/RTS 機能禁止
        MOV.B  #10h,U2C1   ;送受信禁止
        MOV.B  #(80-1),U2BRG ;転送速度 100Kbps
INIT_END:
        RTS

*****
;
;          リードモード処理
*****
;
;
MASTER_IN:
        JSR    MAKE_START ;スタート条件生成
        MOV.B  #0A0h,R0L   ;スレーブアドレス+データ方向ビット(W)
        JSR    OUTPUT_DATA ;スレーブアドレス+データ方向ビット(W)送信
        JNC    MASTER_IN_10
        JSR    ABORT
        FSET   C           ;C=1->再書き込み処理などへ
        RTS

```

MASTER_IN_10:

```

MOV.B #80h,R0L ;ワードアドレス
JSR OUTPUT_DATA ;ワードアドレス送信
JNC MASTER_IN_11
JSR ABORT
FSET C ;C=1->再書き込み処理などへ
RTS

```

MASTER_IN_11:

```

BTST P7_1
JNC MASTER_IN_11 ;SCL(P7_1)が'L'の間待つ
BCLR TE_U2C1 ;送信禁止

BCLR PD7_0 ;SDA(P7_0)入力
MOV.B #00h,U2MR ;シリアル I/O モード無効=ポート制御
JSR CHG_scl_L ;SCL='L'

```

MASTER_IN_12:

```

BTST P7_0
JNC MASTER_IN_12 ;SDA(P7_0)が'L'の間待つ
JSR WAIT_10microSEC

JSR MAKE_START ;スタート条件生成
MOV.B #0A1h,R0L ;スレーブアドレス+データ方向ビット(R)
JSR OUTPUT_DATA ;スレーブアドレス+データ方向ビット(R)送出
JNC MASTER_IN_13
JSR ABORT
FSET C ;C=1->再書き込み処理などへ
RTS

```

MASTER_IN_13:

```

BTST P7_1
JNC MASTER_IN_13 ;SCL(P7_1)が'L'の間待つ
BCLR TE_U2C1 ;送信禁止
JSR INPUT_DATA ;データ入力
JSR MAKE_STOP ;ストップ条件生成
FCLR C
RTS

```

```

*****
;
; 開始(スタート)条件生成
*****
;
;

```

MAKE_START:

```

JSR WAIT_5microSEC
JSR CHG_scl_H ;SCL='H'

```

MAKE_ST1:

```

BTST P7_1
JNC MAKE_ST1 ;SCL(P7_1)端子の入力が'H'になるまで待つ
BTST P7_0
JC MAKE_ST2 ;SDA(P7_0)='H'->MAKE_ST2
JSR CHG_sda_L ;SDA(P7_0)から'L'出力
MOV.B #00h,U2MR ;シリアル I/O 無効=ポート制御
JSR CHG_scl_L ;SCL(P7_1)から'L'出力

```

```

JSR    WAIT_10microSEC
JSR    CHG_sda_H          ;SDA='H'
JSR    WAIT_10microSEC
MAKE_ST2:
JSR    CHG_sda_L          ;スタート条件送出
JSR    WAIT_5microSEC
MOV.B  #02h,U2MR          ;IIC モード設定
RTS

;*****
;
;                停止(ストップ)条件生成
;*****
;
;
MAKE_STOP:
BCLR   TE_U2C1            ;送信禁止
BCLR   RE_U2C1            ;受信禁止
BCLR   PD7_0              ;SDA(P7_0)入力
MOV.B  #00h,U2MR          ;シリアル I/O 無効=ポート制御
JSR    CHG_scl_L          ;SCL(P7_1)'L'出力
JSR    CHG_sda_L          ;SDA(P7_0)'L'出力
JSR    WAIT_5microSEC
JSR    CHG_scl_H          ;SCL(P7_1)入力
MAKE_SP_1:
BTST   P7_1
JNC    MAKE_SP_1          ;SCL(P7_1)に'H'が入力されるまで待つ
JSR    WAIT_10microSEC
JSR    CHG_sda_H          ;ストップ条件送出
RTS

;*****
;
;                データ送信
;*****
;
;
OUTPUT_DATA:
BSET   TE_U2C1            ;送信許可
BCLR   IR_S2TIC           ;アクノリッジ未検出割り込み要求フラグクリア
BCLR   IR_S2RIC           ;アクノリッジ検出割り込み要求フラグクリア
OR.W   #100h,R0           ;SDA 開放ビット('H')付加
MOV.W  R0,U2TB            ;送信データ設定
OUTPUT_1:
BTST   IR_S2RIC           ;アクノリッジ検出割り込み要求フラグ=1?
JNC    OUTPUT_2           ;アクノリッジ未検出割り込み要求フラグの判定へ
JSR    WAIT_10microSEC
FCLR   C
RTS                        ;C=0 のときアクノリッジ検出

OUTPUT_2:
BTST   IR_S2TIC           ;アクノリッジ未検出割り込み要求フラグ=1?
JNC    OUTPUT_1           ;アクノリッジ検出割り込み要求フラグの判定へ
RTS                        ;C=1 のときアクノリッジ未検出

;*****
;
;                データ受信
;*****
;
;

```

```

;
INPUT_DATA:
    BSET    TE_U2C1          ;送信許可
    BSET    RE_U2C1          ;受信許可
    BCLR    IR_S2TIC         ;アクノリッジ未検出割り込み要求フラグクリア
    BCLR    IR_S2RIC         ;アクノリッジ検出割り込み要求フラグクリア
    MOV.W   #1FFh,U2TB      ;アクノリッジ非生成(最終データ)
ID_10:
    BTST    IR_S2TIC
    JC      ID_20
    BTST    IR_S2RIC
    JNC     ID_10
;割り込み要求フラグが'1'になるまで待つ

ID_20:
    MOV.B   U2RB,DATA        ;受信データ格納
    JSR    WAIT_10microSEC
    BCLR    TE_U2C1          ;送信禁止
    BCLR    RE_U2C1          ;受信禁止
    RTS

```

```

*****
;
;                               停止処理 (NACK 時)
;
*****
;
;

```

```

ABORT:
    BCLR    PD7_0            ;SDA(P7_0)入力
    BSET    P7_0             ;SDA(P7_0)'H'
    BCLR    TE_U2C1          ;送信禁止
    MOV.B   #00h,U2MR        ;シリアル I/O 無効=ポート制御
    JSR    WAIT_5microSEC
    JSR    MAKE_STOP         ;ストップ信号送出
    RTS

```

```

*****
;
;                               ポート切り替え処理
;
*****
;
;

```

```

CHG_scl_H:
    BCLR    PD7_1
    RTS
;SCL='H'

```

```

CHG_scl_L:
    BCLR    P7_1
    NOP
    BSET    PD7_1
    RTS
;SCL から'L'出力

```

```

CHG_sda_H:
    BCLR    PD7_0
    RTS
;SDA='H'

```

```

CHG_sda_L:
    BCLR    P7_0
    NOP
    BSET    PD7_0
;SDA から'L'出力

```


RTS

```
*****  
;  
;                                     待ち時間処理  
*****  
;  
;  
WAIT_10microSEC:  
    .MREPEAT    40  
    NOP  
    .ENDR  
  
WAIT_7_5_microSEC:  
    .MREPEAT    40  
    NOP  
    .ENDR  
  
WAIT_5microSEC:  
    .MREPEAT    40  
    NOP  
    .ENDR
```

4. ホームページとサポート窓口

ルネサス テクノロジホームページ
<http://www.renesas.com/>

M16C ファミリ MCU 技術サポート窓口
E-mail: support_apl@renesas.com

改訂記録

Rev.	発行日	改訂内容	
		ページ	ポイント
1.00	2004.02.25	-	初版発行

安全設計に関するお願い

1. 弊社は品質、信頼性の向上に努めておりますが、半導体製品は故障が発生したり、誤動作する場合があります。弊社の半導体製品の故障又は誤動作によって結果として、人身事故、火災事故、社会的損害などを生じさせないような安全性を考慮した冗長設計、延焼対策設計、誤動作防止設計などの安全設計に十分ご注意ください。

本資料ご利用に際しての留意事項

1. 本資料は、お客様が用途に応じた適切なルネサス テクノロジ製品をご購入いただくための参考資料であり、本資料中に記載の技術情報についてルネサス テクノロジが所有する知的財産権その他の権利の実施、使用を許諾するものではありません。
2. 本資料に記載の製品データ、図、表、プログラム、アルゴリズムその他応用回路例の使用に起因する損害、第三者所有の権利に対する侵害に関し、ルネサス テクノロジは責任を負いません。
3. 本資料に記載の製品データ、図、表、プログラム、アルゴリズムその他全ての情報は本資料発行時点のものであり、ルネサス テクノロジは、予告なしに、本資料に記載した製品または仕様を変更することがあります。ルネサス テクノロジ半導体製品のご購入に当たりますは、事前にルネサス テクノロジ、ルネサス販売または特約店へ最新の情報をご確認頂きますとともに、ルネサス テクノロジホームページ(<http://www.renesas.com>)などを通じて公開される情報に常にご注意ください。
4. 本資料に記載した情報は、正確を期すため、慎重に制作したものです。万一本資料の記述誤りに起因する損害がお客様に生じた場合には、ルネサス テクノロジはその責任を負いません。
5. 本資料に記載の製品データ、図、表に示す技術的な内容、プログラム及びアルゴリズムを流用する場合は、技術内容、プログラム、アルゴリズム単位で評価するだけでなく、システム全体で十分に評価し、お客様の責任において適用可否を判断してください。ルネサス テクノロジは、適用可否に対する責任を負いません。
6. 本資料に記載された製品は、人命にかかわるような状況の下で使用される機器あるいはシステムに用いられることを目的として設計、製造されたものではありません。本資料に記載の製品を運輸、移動体用、医療用、航空宇宙用、原子力制御用、海底中継用機器あるいはシステムなど、特殊用途へのご利用をご検討の際には、ルネサス テクノロジ、ルネサス販売または特約店へご照会ください。
7. 本資料の転載、複製については、文書によるルネサス テクノロジの事前の承諾が必要です。
8. 本資料に関し詳細についてのお問い合わせ、その他お気づきの点がございましたらルネサス テクノロジ、ルネサス販売または特約店までご照会ください。